

Материалы для студентов (Леонов Е.А., 2026)

Учебная DSGE-модель:

RBC+ монополистическая конкуренция + жесткость цен + монетарный сектор

Малые греческие буквы - калибруемые параметры

Пусть все агенты i и j равномерно распределены на отрезке $[0,1]$

Задача домохозяйств (j)

Функция ценности

$$U_t(j) = \max_{c,L} \left\{ E_t \sum_{s=0}^{\infty} \beta^s \left(\frac{C_{t+s}(j)^{1-\theta}}{1-\theta} - \phi \frac{(L_{t+s}(j))^{1+\psi}}{1+\psi} \right) \right\}$$

Бюджетное ограничение

$$W_t L_t(j) + R_t K_t(j) + R b_{t-1} B_{t-1}(j) + P r_t(j) = P_t C_t(j) + P_t I_t(j) + B_t(j)$$

Обратите внимание, что здесь у нас включены **однопериодные облигации** $B_t(j)$ с доходностью $1 > (R b_{t-1} - 1) > 0$. Также домохозяйства получают дивиденды от фирм, т.е. полученная фирмами прибыль полностью распределяется домохозяйствам.

Динамика капитала, которым владеют домохозяйства:

$$K_{t+1}(j) = (1 - \delta) K_t(j) + I_t(j)$$

Вспомнить, что здесь есть FOC для агрегированного случая-

Что меняется при включении в бюджетное ограничение облигаций и дивидендов от прибыли?

(ВАЖНО: это легкое задание!!!)

Задача фирм

Фирмы работают в структуре монополистической конкуренции (см. материал семинара по модели Диксита-Стиgliца):

Остаточная функция спроса на продукцию i-фирмы:

$$Y_t(i) = \left(\frac{P_t(i)}{P_t} \right)^{-\sigma} Y_t$$

где

$$Y_t = \left(\int_0^1 Y_t(i)^{\frac{\eta-1}{\eta}} di \right)^{\frac{\eta}{\eta-1}}, \quad P_t = \left(\int_0^1 P_t(i)^{1-\eta} di \right)^{\frac{1}{1-\eta}}$$

Производственная функция Кобба-Дугласа:

$$Y_t(i) = A_t K_t(i)^\alpha L_t(i)^{1-\alpha}$$

С одной стороны фирмы минимизируют расходы, принимая рыночные цены на труд и капитал как заданные, и выбирая оптимальный объем труда:

$$\begin{cases} \min_{L,K} \{W_t L_t(i) + R_t K_t(i)\} \\ \hat{Y}_t(i) = K_t(i)^\alpha L_t(i)^{1-\alpha} \end{cases}$$

Отсюда оптимальный выбор труда и капитала должен удовлетворять правилу:

Не понял как получили эту формулу $\rightarrow \frac{K_t(i)}{L_t(i)} = \frac{W_t}{R_t} \frac{\alpha}{1-\alpha}$

отсюда предельные издержки имеют вид (вспомнить почему):

$$MC_t(i) = MC_t = \frac{R_t^\alpha W_t^{1-\alpha}}{A_t^\alpha (1-\alpha)^{(1-\alpha)}}$$

С другой стороны, фирмы максимизируют прибыль путем выбора оптимальных цен

$$V = E_t \sum_{s=0}^{\infty} Pr_{t+s} = E_t \sum_{s=0}^{\infty} \beta^s \left[(P_{t+s}(i) - MC_{t+s}) Y_{t+s}(i) - \frac{\mu}{2} \left(\frac{P_{t+s}(i)}{P_{t+s-1}(i)} - 1 \right)^2 P_{t+s}(i) Y_{t+s}(i) \right]$$

Здесь $\frac{\mu}{2} \left(\frac{P_{t+s}(i)}{P_{t+s-1}(i)} - 1 \right)^2 P_{t+s}(i) Y_{t+s}(i)$ расходы фирм, связанные с изменением цен, μ – параметр жесткости цен

Вспомнить, что здесь есть ФОС для агрегированного случая

Правило Центробанка (как устанавливается ставка):

$$\frac{Rb_t}{R_{ss}} = M_t \left(\frac{Rb_{t-1}}{R_{ss}} \right)^\gamma \left(\frac{E_t \pi_{t+1}}{\pi_{ss}} \right)^{\tau(1-\gamma)}$$

π_{ss} – целевой уровень инфляции

R_{ss} – равновесная процентная ставка по однопериодным бескупонным облигациям, соответствующая целевому уровню инфляции.

Подумать, почему именно так устроено это правило

Замыкание модели:

$$Y_t = C + I - \frac{\mu}{2} \left(\frac{P_t}{P_{t-1}} - 1 \right)^2 Y_t$$

Т.е. всё что в экономике произведено идет на потребление и инвестиции с учетом тех потерь, которые создает жесткостью цен (ведь эти дополнительные расходы фирм – это по сути потери акционеров, т.е. домохозяйств).

Стохастические процессы:

$$\ln(A_t) = \rho_{TFP} \ln(A_{t-1}) + \varepsilon_{TFP,t} , \quad \varepsilon_{TFP,t} \sim N(0, \sigma_{TFP})$$

$$\ln(M_t) = \rho_{TFP} \ln(M_{t-1}) + \varepsilon_{M,t} , \quad \varepsilon_{M,t} \sim N(0, \sigma_M)$$

Задача: проанализировать код (калибровку, уравнения модели, расчет равновесия).

Какие особенности задачи и решения вы нашли? Как их объяснить?

Листинг расчетного файла

```
var Y K L C I W Wreal R Rreal Rb Rbreal MC MCreall Pr Prreal P pi tfp mon;
varexo tfp_shock mon_shock;
predetermined_variables K;
parameters alpha betta delta theta psi phi nu mu rho_tfp rho_mon sig_tfp sig_mon gamma tau piss
Rss;
    alpha    = 0.35 ;
    betta     = 0.98 ;
    delta     = 0.025;
    theta     = 1.25 ;
    psi       = 0.5  ;
    phi       = 16.0 ;
    nu        = 6.0  ;
    mu        = 30.0 ;
    rho_tfp   = 0.2  ;
    rho_mon   = 0.0  ;
    sig_tfp   = 0.001 ;
    gamma     = 0.8  ;
    tau       = 1.5  ;
    piss     = 1.0;
    Rss       = 1/betta;
    sig_mon   = 0.0025 ;

model;
    (C(+1)/C)^theta - betta*(R(+1)/P(+1)+1-delta) = 0;
    (C(+1)/C)^theta - betta*(Rb/pi(+1))=0;
    K(+1) - (1-delta)*K - I = 0;
    phi*(L^psi)*(C^theta) - W/P = 0;
    Y - tfp*K^alpha*L^(1-alpha) = 0;
    K/L - (W/R)*(alpha/(1-alpha)) = 0;
    MC - (1/tafp)*(W/(1-alpha))^(1-alpha)*(R/alpha)^(alpha)=0;
    (1-nu)+nu*MC/P - mu*(P/P(-1) - 1)*(P/P(-1)) + betta*mu*(P(+1)/P - 1)*(P(+1)/P)^2*(Y(+1)/Y) = 0;
    Pr = (P-MC)*Y - (mu/2)*(P/P(-1)-1)^2*P*Y ;
    Y - C - I - (mu/2)*(P/P(-1)-1)^2*Y = 0;
    pi = P/P(-1);
    log(tfp) = rho_tfp*log(tfp(-1))+tfp_shock;
    (Rb/Rss) = mon*( (Rb(-1)/Rss)^gamma*(pi(+1)/piss)^(tau*(1-gamma)));
    log(mon) = rho_mon*log(mon(-1)) + mon_shock;
    Wreal = W/P; Rreal = R/P; Rbreal= Rb/P; Prreal= Pr/P; MCreall= MC/P;
end;

steady_state_model;
    Rb=1/betta;
    mon=1;
    tfp=1;
    MC = 1; pi = 1;
    P = MC*nu/(nu-1);
    Rb = 1/betta;
    R = P*(1/betta + delta - 1);
    W = (MC*((1-alpha)^(1-alpha)*alpha^alpha)/(R^alpha))^(1/(1-alpha));
    K2L= (W/R)*alpha/(1-alpha);
    C2L= K2L^alpha-delta*K2L;
    L = ((W/P/phi)/(C2L^theta))^(1/(psi+theta));
    K = K2L*L;
    C = C2L*L;
    Y = K^alpha*L^(1-alpha);
    I = delta*K;
    Pr=(P - MC)*Y;
    Wreal = W/P; Rreal = R/P; Rbreal= Rb/P; Prreal= Pr/P; MCreall= MC/P;
end;
steady;

shocks;
var tfp_shock=sig_tfp^2;
var mon_shock=sig_mon^2;
end;

stoch_simul;
```